```
(* Workspace setup *)

homedir = NotebookDirectory[];

savedir := homedir <> "saves/";

oct1pts := savedir <> "oct1pts.wl"

tstamp := ToString@UnixTime[];
```

```
(* Selection and component parameters *)

keepedge[x_, xes_, y_] :=
  If[(Norm[y - x] < Total@(edgelen /@ xes)) && (Length@xes < maxconn), True, False];

within = 0.0002; (* Maximum vertex distance from origin *)

maxconn = 5; (* Max connections per vertex *)

maxedgelen = 0.00007; (* Maximum allowed edge length *)

{lv, uv} = {1, 7}; (* Vertex number interpolation bounds *)

{lr, ur} = {0.0000015, 0.0000035}; (* Min/max radii of spheres/catenary endpts *)

{la, ua} = {0.3, 0.1};
(* Min/max catenary curvature (smaller number → more curvy) *)

na = 20; (* Catenary subdivision count *)
```

```
(* Vertex selection from dataset *)

octpts = Import[oct1pts];

inpts = Select[octpts, (Norm[#] < within) &];

g = NearestNeighborGraph[inpts, {maxconn, maxedgelen}];

Print[If[ConnectedGraphQ@g,
   "Connected graph with " <> ToString[Length@VertexList@g] <> " vertices (" <>
    ToString@N[100 Length@VertexList@g / Length@octpts, 2] <> "%)", "Not Connected"]];
```

```
Connected graph with 248 vertices (0.65%)
```

In[•]:=
```
(* Interpolation and other helper functions *)
```

In[•]:=
```
edgelen := Norm[(# // Normal)[[1]] - (# // Normal)[[2]]] &;
```

In[•]:=
```
{le, ue} = MinMax@(edgelen /@ (EdgeList@g));
```

In[•]:=
```
lerp[v1_, v2_, n_] := Table[(1 - t) v1 + t v2, {t, Range[0, 1, 1 / (n - 1)]}];
(* linspace equivalent *)
```

In[•]:=
```
linmap[x_, a_, b_, c_, d_] := c + (x - a)/(b - a) (d - c);
(* linearly map a value x∈(a,b) to range (c,d) *)
```

In[•]:=
```
vrmap[vert_] := Norm[vert]/within linmap[VertexDegree[g, vert], lv, uv, lr, ur];
(* Map vertex degree to a radius *)
```

In[•]:=
```
amap[v1_, v2_] := linmap[Norm[v2 - v1], le, ue, la, ua];
(* Map distance between vertices to a curvature value *)
```

In[•]:=
```
caterp[x1_, x2_, n_, a_] := Table[
    linmap[t, 0, 1, vrmap[x1], vrmap[x2]] Cosh[(t - 1/2)/a] / Cosh[-1/2/a], {t, Range[0, 1, 1 / (n - 1)]}];
(* Catenary radii between two vertices *)
```

In[•]:=
```
(* Generate and save graphic *)
```

In[•]:=
```
tube := Tube[lerp[#[[1]], #[[2]], na], caterp[#[[1]], #[[2]], na, amap[#[[1]], #[[2]]]]] &;
```

In[•]:=
```
sphere := Sphere[#, vrmap@#] &;
```

```
gpl = GraphPlot3D[g, BoxRatios → {1, 1, 1},
  VertexShapeFunction → sphere, EdgeShapeFunction → tube, ImageSize → 800,
  Background → Darker[Gray, 0.8], VertexStyle → Darker[LightBlue, 0.1],
  Lighting → "ThreePoint", EdgeStyle → Darker[LightBlue, 0.1]]
```

Out[•]=





```
ts = tstamp;
Export[savedir <> "img_" <> ts <> ".png", gpl];
Export[savedir <> "model_" <> ts <> ".stl", gpl, "STL"];
```

```
(* Record parameters *)

report = ((Table[{#[[i]], (ToExpression[#[[i]]])}, {i, Length@#}] &) @
     {"ts", "within", "maxconn", "maxedgelen", "lv",
      "uv", "lr", "ur", "la", "ua", "na"}) // TableForm;
Export[savedir <> "params_" <> ts <> ".csv", report, "CSV"];
```